
Overview

The purpose of this script is to serve as a basic analysis for the general geometry of the tiles to be used for the NSTX-CSU. It is by no means a thorough analysis of the tile geometry or mounting system and is therefore subject to error. It is used primarily to find the worst case given a general scenario (i.e., over-constraining a tile and finding the worst case stresses that may develop due to thermal expansion, eddy currents, and halo currents). It may also be used to find the locations on a tile that may present a problem in a given situation. The script will perform eddy current, halo current, and thermal analyses on the tile and plot the von Mises stress for the combined loads upon completion.

The tile geometry that is used in this script is a basic block with a T-slot running the length of the tile, centered horizontally. Several sample images (with dimension labels) are below.

	[image: image1.emf]
Figure 1. General overview of the tile geometry including thickness, width, and height dimensions.

Figure 2. Top view of the tile geometry including t1, t2, t3, and t4 dimensions, which are used to size the T-slot.


The geometry used in the ANSYS model is susceptible to high corner stresses, which can generally be ignored as the true tile geometry will include chamfers and fillets to avoid these stresses. This is assumed based on the way in which the stresses appear to develop and has been demonstrated in other models where similar stresses have emerged. The stresses are calculated by reading in forces from each of the individual analysis types and specifying constraints.

How to Use This Script

There are several files that are required to be in the working directory in order to successfully run this script: ‘BasicTile.txt’, ‘input.txt’, ‘constraints.txt’, and ‘b_mac.mac’. ‘BasicTile.txt’ is the actual script. It reads table data from ‘input.txt’ and the constraint information from ‘constraints.txt’. If these files are missing or improperly formatted, the script may generate unexpected results. To change the input parameters used for the tile analysis, the user simply needs to change the numbers that are stored in the second column of ‘input.txt’. The first column is the indexing column and never needs to be modified unless the structure of ‘BasicTile.txt’ is changed. There are eighteen rows and two columns. The variable that is stored in each row is below. There is also a comment at the bottom of the file describing each of the variables.
	Index
	Variable
	Index
	Variable

	1
	Material type
	10
	Horizontal Bdot (in T/s)

	2
	Tile width (in inches)
	11
	Vertical Bdot (in T/s)

	3
	Tile height (in inches)
	12
	Normal Bdot (in T/s)

	4
	Tile Thickness (in inches)
	13
	Horizontal B (in Tesla)

	5
	T1 (in inches)
	14
	Vertical B (in Tesla)

	6
	T2 (in inches)
	15
	Normal B( in Tesla)

	7
	T3 (in inches)
	16
	Horizontal halo current density (in A/m2)

	8
	T4 (in inches)
	17
	Vertical halo current density (in A/m2)

	9
	Heat flux (in W/m2)
	18
	Normal halo current density (in A/m2)


There are eight material types to choose from. They are specified by their index numbers which are listed below.

	Index number
	Material

	1
	ATJ Graphite

	2
	Poco AXF-5Q Graphite

	3
	Sigrabond CFC 1501G (2D weave)

	4
	Isostatic Graphite R*510 (unknown source)

	5
	Annealed Molybdenum

	6
	Annealed Tungsten

	7
	Thermagard (2D weave CFC)

	8
	Meggagard (3D weave CFC)


The ‘constraints.txt’ file is constructed in an entirely different manner. The user must input the commands to be used for the physical constraints on the tile. The reason that a more user-friendly approach wasn’t developed is that the results are very sensitive to the constraints, so the constraints may need to be finely modified to achieve a reasonable result. For most analyses, the tile is over-constrained in order to find the worst case stresses. High stresses at locations of nodal constraints are generally ignored. The high stresses develop because a small number of nodes are constrained to represent the way in which a piece of hardware (such as a T-bar or pin) will hold the tile. The small number of nodes that are chosen develop unrealistically high stresses.  Calculating the bearing stress for the true geometry based on the reaction forces from the analysis can tell you whether or not the assumption is correct in a particular case. The reaction force on a group of constrained nodes is independent (at least for small scale changes) of the number of nodes chosen.
After the entire script has run once, the stress solution may be repeatedly run until you are confident with the results. To do this, modify the ‘constraints.txt’ file, clear all of the current constraints, use the command ‘/input,constraints,txt’, and then use the command ‘solve’. This will simply re-run the solution using different physical constraints.

The force types used may also be selected by deleting all forces and body forces then loading in the desired forces using the ‘ldread’ command.

‘b_mac.mac’ is simply used to impose a uniform vector potential on all of the selected nodes.

BasicTile Script

The BasicTile script is broken up into several parts, each separated by a string of exclamation marks and a descriptive title.

Tile Parameters

First the tile parameters are stored. These parameters are taken from the file ‘input.txt’ and stored as variables for later use. The file ‘input.txt’ is read into ANSYS as a table and each component of the table is then stored.

Build Tile

The tile is built by creating a block that is lying on its side, which then has two other blocks subtracted from it to create a T-slot. The tile is then rotated so the vertical direction is the y-direction, the horizontal direction is the x-direction, and the direction normal to the tile face is the z-direction. The global origin is centered parallel to the back surface of the tile.

Material Properties

The material properties are chosen based on the user’s selection in the file ‘input.txt’. The material options are listed above in the section called ‘How to Use This Script’. The selection must be input as an integer from one to eight. If any other number or variable is chosen, the script will terminate with a message to the user.
Mesh

The script meshes the model with bricks using SOLID97 where keyopt(1) = 1. This is an 8-node electromagnetic brick type element. This was chosen for the ease of applying vector potentials and because it has easy-to-use thermal and structural counterparts. The default element size is one-thirtieth of the width of the tile. This can be changed if the calculation is taking too long or if there are not enough elements through the thickness of the tile. There should be a minimum of three elements between any two sections of open space. To verify that there are enough elements, it may be desired to run the model at several mesh densities to verify that the solution has converged.
Eddy Current Analysis

This section performs a transient electromagnetic analysis where the change in magnetic field per time and the final magnetic field are chosen by the user. The resistive solution is used because it should be the worst case scenario. The reason that the resistive solution is the worst case solution is that the eddy currents are fully developed, regardless of how large the object of interest is. This ignores the amount of time that it takes for the currents to be fully induced and changes the current directly from zero to the full value. Since the eddy currents will be at their largest magnitudes and the background magnetic field is the same regardless of which solution is chosen, the nodal forces will also be the largest. A single node is constrained to zero volts as a reference for ANSYS. This point may be arbitrary because voltage is a relative measurement rather than an exact measurement and constraining a single point to zero volts simply acts as reference point rather than as a ground. If multiple points are constrained to zero volts, the constraints can cause the currents to unexpectedly change direction and create unexpected results. The analytical results for the eddy currents in a rectangular loop are comparable to the ANSYS solution. A sample comparison can be seen in the ‘Analytical Comparisons’ section of this report. Upon completion of this portion, the eddy currents are briefly plotted for the user to see.
Halo Current Analysis

For this portion of the script, a background magnetic field and a current density are imposed on all of the nodes in the tile. The result of this portion of the script will be nodal forces which are formed by crossing the currents with the magnetic fields.
Thermal Analysis

For the thermal analysis, a user-specified heat flux is applied to the front surface of the tile for five seconds, as that is the pulse length to be used following the NSTX upgrade. The initial temperature of the body is set to 25 °C. The approximate average change in temperature of the ANSYS model can be compared to the analytical average change by adding the temperatures of all of the nodes and dividing by the number of nodes. This is not an exact solution since it does not include space between nodes, but it can be used as simple check. For most analyses, the error is less than three percent using this method. The derivation of the analytical solution can be seen in the ‘Analytical Comparisons’ section of this report. This analysis is performed using no cooling of any type. For this analysis, the element type is changed to SOLID70.

Structural Analysis

The structural analysis combines the results from all of the previous analyses. It is also a good idea to look at the stresses created by each individual analysis. For the constraints, the ‘constraints.txt’ file is used. After the problem is solved, the script plots the stress in the x-direction. This is normally (but not always) the largest magnitude stress. The reason this one is plotted is that the gives the user a general sense of which sections of the tile are in compression and which are in tension. The constraints can be easily changed to try different mounting methods or analysis types by modifying the ‘constraints.txt’ file, deleting the original constraints, and re-solving the problem. The forces used for the analysis are stored in separate files, so they may be removed from the model and the individual forces from each analysis type can be used. To do this, use the following commands:

/solu

allsel

fdele,all,all

bfdele,all,all

For eddy current stresses: ldread,forc,last,,,,%jobid%_em,rst

For halo current stresses: ldread,forc,last,,,,%jobid%_halo,rmg

For thermal stresses: ldread,temp,last,,,,%jobid%_therm,rth

solve

Analytical Comparisons

These derivations use basic laws of physics and will contain little explanation, though they should be very simple to follow along with.
Eddy Current Solution

The analytical eddy current solution combines Ohm’s Law with Faraday’s Law and uses the geometry of a rectangle for the shape-dependent portions of the solution. The table below describes each of the variables used. The derivation below assumes that there is a uniform change in magnetic field per time and that the field change is normal to the area of the loop. Also, the effects of current concentrations are ignored.
	Variable
	Description

	Magnetic flux

	Time

	Electrical Current

	Electrical Resistance

	Electrical Resistivity

	Length of perimeter of loop

	Cross-sectional area of infinitesimally thin loop

	Change in magnetic field per time

	[image: image10.png]Ajoop




	Area enclosed in loop

	Width of tile

	Height of tile

	Current density

For a rectangle:

The actual current density at the edges of the rectangle will be larger than expected due to the current concentrations in those areas. This equation can only be used as an estimate. A test comparison can be seen below.
	h = 4 inches
	

	w = 4 inches

 = 350 T/s
	

	ρ = 10-5 Ω∙m
	

	ANSYS Solution
	Analytical Solution

	1.0∙106A/m2
	8.9∙105 A/m2

	
	


The sample above has approximately an 11% error. This error can mostly be attributed to current concentrations around the borders of the tile.

Average Temperature
This is the exact analytical solution for the average temperature change in the tile. This solution is generally within 3% of the ANSYS solution. The errors are mainly due to discretization errors. A list of variables and descriptions may also be seen below. The derivation is simply a manipulation of the basic heat formula.

	Variable
	Description

	Heat into or out of tile

	Specific heat

	Mass

	Change in temperature

	Heat flux applied to front surface of tile

	Width of tile

	Height of tile

	Volume of tile

	Thickness of tile

	See images in ‘Overview’ section.

	See images in ‘Overview’ section.

	See images in ‘Overview’ section.

	See images in ‘Overview’ section.

After simplifying:

Required Scripts and Files

The following sections contain samples of all of the required scripts and files. For the case-specific files, the samples are taken from the center stack analyses.

‘BasicTile.txt’

fini

/clear

jobid = 'tile'

/prep7

*dim,vars,table,18,1,1,X,,,0

*tread,vars,input,txt

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Tile Parameters

!

w = vars(2)*.0254

h = vars(3)*.0254

t = vars(4)*.0254

t1 = vars(5)*.0254

t2 = vars(6)*.0254

t3 = vars(7)*.0254

t4 = vars(8)*.0254

! Pulse Parameters

hflx = vars(9)

tem0 = 25.

tpulse = 5.

! Disruption Parameters

!

bdotx=vars(10)
! Across face horizontally

bdoty=vars(11)
! Across face vertically

bdotz=vars(12)
! Normal to front face

! Field at end of Ramp

bx=vars(13)
! Across face horizontally

by=vars(14)
! Across face vertically

bz=vars(15)
! Normal to front face

! Halo Currents 

!

Halox=vars(16)

Haloy=vars(17)

Haloz=vars(18)

!

! Ramp Time

tim=.001  

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Build Tile

!

blc4,-w/2,0,w,t,-h

blc4,-t2/2,0,t2,t3,-h

blc4,-t1/2,t3,t1,t4,-h

vsbv,1,all

numcmp,all

csys,6

vgen,,1,,,0,90,0,,,1

csys,0

vgen,,1,,,0,-h/2,0,,,1

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Material Properties

!

!

mattype = vars(1)

*IF,MATTYPE,EQ,1,THEN

!

! ATJ Graphite

!

MP,DENS,1,1760

MP,MURX,1,1.

MP,RSVX,1,11.7E-6

!

! Structural Prop

MP,EX,1,9.7E9

MP,ALPX,1,3.0E-6

MP,REFT,1,25

MP,NUXY,1,.3

!

! Thermal Properties 

MPTEMP,1,0.,227.,477.,727.,977.,1227

MPTEMP,7,1977.,3227.

MPDATA,C,1,1,600.0,1275.,1600.,1760.,1875.,1940.

MPDATA,C,1,7,2050.,2100.

MPTEMP,1,20.,250.,500.,750.,1000.,1250.

MPTEMP,7,1500.,1750.

MPDATA,KXX,1,1,117.,95.,75.,60.,50.,44.

MPDATA,KXX,1,7,43.,47.

*ELSEIF,MATTYPE,EQ,2,THEN

!

! Poco AXF-5Q Graphite

!

MP,DENS,1,1780

MP,MURX,1,1.

MP,RSVX,1,14.7E-6

!

! Structural Prop

MP,EX,1,11.E9

MP,ALPX,1,7.90E-6

MP,REFT,1,25

MP,NUXY,1,.3

!

! Thermal Properties 

MP,C,1,1000. ! Value approximated based on similar materials

MPTEMP,1,20.,250.,500.,750.,1000.,1250.

MPTEMP,7,1500.,1750.

MPDATA,KXX,1,1,78.,64.,52.,45.,40.,37.

MPDATA,KXX,1,7,35.,34.

*ELSEIF,MATTYPE,EQ,3,THEN

!

! CFC 1501 G

!

MP,REFT,1,25

MP,MURX,1,1.

MP,DENS,1,1430.

MPTEMP,1,20,200,400,600,800,1000

MPTEMP,7,1200,1400,1600,1800,2000

MPDATA,RSVX,1,1,2.56E-5,2.30E-5,2.05E-5,1.84E-5,1.66E-5,1.52E-5

MPDATA,RSVX,1,7,1.42E-5,1.33E-5,1.27E-5,1.21E-5,1.17E-5

!

! Thermal Properties 

MPDATA,C,1,1, 665,1040,1385,1656,1831,1939

MPDATA,C,1,7,1993,2026,2043,2055,2068

MPDATA,KZZ,1,1,4.44,4.58,4.78,4.85,5.05,5.19

MPDATA,KZZ,1,7,5.46,5.67,6.08,6.62,7.44

MPDATA,KXX,1,1,24.4,27.0,28.9,29.5,29.1,28.4

MPDATA,KXX,1,7,27.4,26.8,26.9,27.4,28.4

MPDATA,KYY,1,1,24.4,27.0,28.9,29.5,29.1,28.4

MPDATA,KYY,1,7,27.4,26.8,26.9,27.4,28.4

!

! Structural Prop

MP,EX,1,75.E9

MPTEMP

MPTEMP,1,200,400,600,800,1000

MPTEMP,6,1200,1400,1600,1800,2000

MPDATA,ALPZ,1,1,7.37E-6,7.68E-6,7.96E-6,8.21E-6,8.47E-6

MPDATA,ALPZ,1,6,8.77E-6,9.00E-6,9.31E-6,9.56E-6,9.84E-6

MPDATA,ALPX,1,1,-4.51E-7,-2.18E-7,6.15E-8,3.18E-7,5.27E-7

MPDATA,ALPX,1,6, 8.07E-7, 1.02E-6,1.27E-6,1.53E-6,1.76E-6

MPDATA,ALPY,1,1,-4.51E-7,-2.18E-7,6.15E-8,3.18E-7,5.27E-7

MPDATA,ALPY,1,6, 8.07E-7, 1.02E-6,1.27E-6,1.53E-6,1.76E-6

MP,NUXY,1,.3

*ELSEIF,MATTYPE,EQ,4,THEN

!

! Isostatic Graphite R*510

!

MP,reft,1,25

MP,MURX,1,1.

MPTEMP,1,20.,250.,500.,750.,1000.,1250.

MPTEMP,7,1500.,1750.,2000.

MPDATA,DENS,1,1,1831.,1826.,1819.,1811.,1804.,1795.

MPDATA,DENS,1,7,1786.,1777.,1768.

MPDATA,RSVX,1,1,12.8E-6,10.7E-6,9.11E-6,8.56E-6,8.61E-6,8.87E-6

MPDATA,RSVX,1,7,9.14E-6,9.44E-6,9.97E-6

!

! Structural Prop

MP,EX,1,11.4E9

MPDATA,ALPX,1,1,3.64E-5,4.12E-5,4.53E-5,4.84E-5,5.10E-5,5.31E-5

MPDATA,ALPX,1,7,5.51E-5,5.68E-5,5.85E-5

MP,NUXY,1,.3

!

! Thermal Properties 

MP,C,1,1000. ! Value approximated based on similar materials

MPDATA,KXX,1,1,107.4,86.2,70.2,59.3,52.2,47.5

MPDATA,KXX,1,7,44.6,42.9,43.2

*ELSEIF,MATTYPE,EQ,5,THEN

!

! Annealed Molybdenum

!

MP,reft,1,25

MP,MURX,1,1.

MP,DENS,1,10220.

MPTEMP,1,0,27,727,927,1127,1327

MPTEMP,7,1527,1727,1927,2127,2227,2327

MPTEMP,13,2527,2622

MPDATA,RSVX,1,1, 5.20E-8,5.70E-8,2.39E-7,2.92E-7,3.52E-7,4.12E-7

MPDATA,RSVX,1,7, 4.72E-7,5.35E-7,5.95E-7,6.60E-7,6.92E-7,7.18E-7

MPDATA,RSVX,1,13,7.82E-7,8.14E-7

!

! Structural Prop

MPTEMP

MPTEMP,1,20,1000

MPDATA,EX,1,1,325E9,275E9

MPDATA,NUXY,1,1,.3,.3

MPTEMP,1,27,227,327,527,1127

MPDATA,ALPX,1,1,5.35E-6,6.0E-6,6.0E-6,6.0E-6,6.5E-6

!

! Thermal Properties 

mptemp,1,27,127,327,427,526,627  ! temps converted to deg C 

mptemp,7,727,927,1227,1327,1627,1727

mptemp,13,1827,2227,2327

mpdata,c,1,1,272,271,272,274,276,280

mpdata,c,1,7,285,295,318,326,362,372

mpdata,c,1,13,387,446,462

mptemp

MPTEMP,1,1,27,127,227,327,527,727

MPTEMP,7,927,1127

MPDATA,KXX,1,1,138.,134.,130.,126.,118.,112.

MPDATA,KXX,1,7,105.,100.

*ELSEIF,MATTYPE,EQ,6,THEN

!

! Annealed Tungsten

!

MP,REFT,1,25.

MP,MURX,1,1.

MP,DENS,1,19250.

MPTEMP,1,20,100,500,700,1000,1100

MPTEMP,7,1500,1800,2100,3000

MPDATA,RSVX,1,1, 5.50e-8,7.28e-8,1.80e-7,2.24e-7,3.30e-7,3.47e-7

MPDATA,RSVX,1,7, 4.97e-7,5.75e-7,6.96e-7,1.00e-6

!

! Structural Prop

MPDATA,EX,1,1,411E9

MPDATA,NUXY,1,.28

MP,ALPX,1,4.50E-6

!

! Thermal Properties 

MP,C,1,134.

MPTEMP,1,20,100,500,927,1127,1327

MPTEMP,7,1527,1727,2000

MPDATA,KXX,1,1, 167.4,159.0,121.3,115.1,112.1,108.8

MPDATA,KXX,1,7, 105.9,102.5,92.9

*ELSEIF,MATTYPE,EQ,7,THEN

!

! Thermagard (2-D weave CFC)

!

MP,REFT,1,25

MP,MURX,1,1.

MP,DENS,1,1800.

MP,RSVX,1,1.0e-5

! Structural Properties

MP,EX,1,75.0E9

MP,NUXY,1,0.3

! Thermal Properties

MPTEMP,1,200,1000

MPDATA,KXX,1,1,210,100

MPDATA,KYY,1,1,210,100

MPDATA,KZZ,1,1,50,25

MP,C,1,702

MP,ALPX,1,3.0E-6

MP,ALPY,1,3.0E-6

MP,ALPZ,1,15.0E-6

*ELSEIF,MATTYPE,EQ,8,THEN

!

! Meggagard (3-D weave CFC)

!

MP,REFT,1,25

MP,MURX,1,1.

MP,DENS,1,1800.

MP,RSVX,1,1.0e-5

! Structural Properties

MP,EX,1,75.0E9

MP,NUXY,1,0.3

! Thermal Properties

MP,KXX,1,55

MP,KYY,1,100

MP,KZZ,1,260

MP,C,1,702

MP,ALPX,1,5.0E-6

MP,ALPY,1,3.0E-6

MP,ALPZ,1,2.0E-6

*ELSE

*MSG,ui

No material chosen. Script terminated.

*return,0

*endif

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Mesh

!

et,1,97,1

esize,w/30

vsweep,1

save

fini

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Eddy Current Analysis

!

/filname,%jobid%_em

/solu

antype,trans,new

timint,off

time,1

!

bx0 = bx - bdotx*tim

by0 = by - bdoty*tim

bz0 = bz - bdotz*tim

!

allsel

b_mac,bx0,by0,bz0

solve

!

timint,on

time,1+tim

nsubst,10

outres,all,all

autots,off

b_mac,bx,by,bz

d,1,volt,0

solve

fini

/post1

set,last

plvect,jt,,,,vect,,1

fini

save

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Halo Current Analysis

!

/filnam,%jobid%_halo

/prep7

et,1,97,0

fini

!

/solu

antype,stat,new

nsubst,1

time,1

allsel

ddel,all,all

!

! apply background field

allsel

b_mac,bx,by,bz

!

! apply halo current

esel,s,mat,,1

nsle

bfedel,all,all

bfe,all,js,1,Halox,Haloy,Haloz

!

allsel

solve

fini

!

save

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Thermal Analysis

!

/filname,%jobid%_therm

/prep7

et,1,70

fini

/solu

antype,trans,new

time,tpulse

autots,off

nsubst,10

!

nsel,s,loc,z,t

sf,all,hflux,hflx

allsel

tunif,tem0

!

solve

fini

!

/post1

plnsol,temp

save

fini

!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!!

! Structural Analysis

!

/filname,%jobid%_struct

!

/prep7

et,1,45

fini

!

/solu

antype,static,new

fdel,all,all

fcum,add

ldread,forc,last,,,,%jobid%_em,rst

ldread,forc,last,,,,%jobid%_halo,rmg

ldread,temp,last,,,,%jobid%_therm,rth

/input,constraints,txt

solve

fini

!

/post1

plnsol,sx

save
‘b_mac.mac’

!

! macro to set magnetic vector potential (in rectangular) to produce uniform B field

! usage:  b_mac,bx,by,bz

!

*get,n,node,,num,min

*get,nnode,node,,count

*do,i,1,nnode

d,n,ax,(arg2*nz(n)-arg3*ny(n))/2.

d,n,ay,(arg3*nx(n)-arg1*nz(n))/2.

d,n,az,(arg1*ny(n)-arg2*nx(n))/2.

*get,n,node,n,nxth

*enddo

Sample ‘input.txt’ File

1
7

2
4.258

3
4.462

4
1.0

5
1.14

6
0.755

7
0.31

8
0.14

9
1.53e6

10
0.

11
950.

12
350.

13
3.0

14
0.45

15
-0.35

16
0.

17
7.76e6

18
2.00e6

! NSTX Data File

! 1 = Material Type

! 2 = Width
[inches]

! 3 = Height
[inches]

! 4 = Thickness
[inches]

! 5 = T1
[inches]

! 6 = T2
[inches]

! 7 = T3
[inches]

! 8 = T4
[inches]

! 9 = Heat flux [W/m^2]

! 10 = bdot in x direction (across face horizontally) [T/s]

! 11 = bdot in y direction (across face vertically) [T/s]

! 12 = bdot in z direction (normal to face) [T/s]

! 13 = B field across face horizontally [T/s]

! 14 = B field across face vertically [T/s]

! 15 = B field normal to face [T/s]

! 16 = halo current density across face horizontally [A/m^2]

! 17 = halo current density across face vertically [A/m^2]

! 18 = halo current density normal to face [A/m^2]

Sample ‘constraints.txt’ File

allsel

ddele,all,all

! T-slot and back surface radial constraints

nsel,s,loc,x,-t1/2,t1/2

nsel,r,loc,y,-1.331*.0254,1.331*.0254

nsel,r,loc,z,t3

nsel,a,loc,z,0

d,all,uz,0

! T-slot toroidal constraints

nsel,s,loc,x,-t2/2

nsel,a,loc,x,t2/2

nsel,r,loc,y,-1.331*.0254,1.331*.0254

nsel,r,loc,z,0,0.4e-2

d,all,ux,0

! Locating Pin

nsel,s,loc,x,-0.0025,0.0025

nsel,r,loc,y,-0.0025,0.0025

nsel,r,loc,z,t3+t4

d,all,uy,0

allsel
